



TSwap Protocol Audit Report

Version 1.0

Tanu Gupta

July 22, 2025

TSwap Protocol Audit Report

Tanu Gupta

July 22, 2025

Prepared by: Tanu Gupta

Lead Security Researcher:

- Tanu Gupta

Table of Contents

- Table of Contents
- Protocol Summary
- TSwap
 - TSwap Pools
 - Core Invariant - $x * y = k$
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
 - High

- * [H-1] Incorrect fee calculation in `TSwapPool::getInputAmountBasedOnOutput` causes protocol to take too many tokens from the user, resulting in a loss of funds
- * [H-2] Lack of slippage protection in `TSwapPool::swapExactOutput` causes users to spend way more tokens
- * [H-3] `TSwapPool::sellPoolTokens` mismatches input and output tokens causing users to receive incorrect amount of tokens
- * [H-4] In `TSwapPool::_swap` the extra tokens given to users after every `swapCount` reaches 10 or more, breaks the invariant of $x * y = k$
- Medium
 - * [M-1] `TSwapPool::deposit` is missing a `deadline` parameter check causing transactions to be processed after the deadline
 - * [M-2] Rebase, fee-on-transfer and ERC-777 break protocol invariant $x * y = k$
- Low
 - * [L-1] The arguments set for this event `TSwapPool::LiquidityAdded` are not set in correct order
 - * [L-2] Default value returned by `TSwapPool::swapExactInput` results in incorrect value being returned
- Informational
 - * [I-1] Unused custom error `PoolFactory::PoolFactory__PoolDoesNotExist` in `PoolFactory` contract
 - * [I-2] `wethToken` parameter in the constructor of the `PoolFactory` contract is not checked for zero address
 - * [I-3] `liquidityTokenSymbol` variable of `PoolFactory::createPool` function should use `symbol()` instead of `name()`
 - * [I-4] `deadline` parameter in `TSwapPool::deposit` function is not used
 - * [I-5]: Event is missing `indexed` fields
 - * [I-6] `TSwapPool::TSwapPool__WethDepositAmountTooLow` in this event, `TSwapPool::MINIMUM_WETH_LIQUIDITY` is not required to be emitted as this a constant value
 - * [I-7] `TSwapPool::deposit` function is not following CEI for the first-time liquidity deposit
 - * [I-8] Usage of magic numbers such as 997 and 1000 is not recommended
 - * [I-9] Natspec is not defined for the function `TSwapPool::swapExactInput`
 - * [I-10] `TSwapPool::swapExactOutput` natspec is missing the `@param` tag for `deadline`
- Gas
 - * [G-1] `TSwapPool::swapExactInput` and `TSwapPool::totalLiquidityTokenSupply`

functions should be marked as `external` instead of `public`

Protocol Summary

TSwap

This project is meant to be a permissionless way for users to swap assets between each other at a fair price. You can think of T-Swap as a decentralized asset/token exchange (DEX). T-Swap is known as an Automated Market Maker (AMM) because it doesn't use a normal "order book" style exchange, instead it uses "Pools" of an asset. It is similar to Uniswap.

TSwap Pools

The protocol starts as simply a `PoolFactory` contract. This contract is used to create new "pools" of tokens. It helps make sure every pool token uses the correct logic. But all the magic is in each `TSwapPool` contract.

Every pool is a pair of `TOKEN X` & `WETH`.

There are 2 functions users can call to swap tokens in the pool.

- `swapExactInput`
- `swapExactOutput`

Core Invariant - $x * y = k$

- x = Token Balance X
- y = Token Balance Y
- k = The constant ratio between X & Y

This codebase is based loosely on Uniswap v1

Disclaimer

The team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings described in this document correspond the following github repository t-swap

Scope

- Commit Hash: e643a8d4c2c802490976b538dd009b351b1c8dda
- In Scope:

```
1 ./src/  
2 #-- PoolFactory.sol  
3 #-- TSwapPool.sol
```

- Solc Version: 0.8.20
- Chain(s) to deploy contract to: Ethereum
- Tokens:
 - Any ERC20 token

Roles

- Liquidity Providers: Users who have liquidity deposited into the pools. Their shares are represented by the LP ERC20 tokens. They gain a 0.3% fee every time a swap is made.
- Users: Users who want to swap tokens.

Executive Summary

Found the bugs using a tool called foundry including invariant fuzzing.

Issues found

Severity	Number of issues found
High	4
Medium	2
Low	2
Info	10
Gas	1
Total	19

Findings

High

[H-1] Incorrect fee calculation in `TSwapPool::getInputAmountBasedOnOutput` causes protocol to take too many tokens from the user, resulting in a loss of funds

Description: The `getInputAmountBasedOnOutput` function in the `TSwapPool` contract calculates the input amount based on the output amount and reserves, but it uses a fee calculation that is not aligned with the expected behavior. When calculating the fee, it scales the amount by 10_000 instead of 1000.

Impact: The fee is applied incorrectly, leading to the protocol taking more tokens from the user than intended. **Proof of code**

```
1 function test_UserIsChargedMoreThanIntended() external {
2     uint256 initialWethAmount = 100e18;
3     uint256 initialPoolTokenAmount = 100e18;
4     //initial liquidity added to the pool
5     vm.startPrank(liquidityProvider);
6     weth.approve(address(pool), type(uint256).max);
7     poolToken.approve(address(pool), type(uint256).max);
```

```
8      pool.deposit(initialWethAmount, 0, initialPoolTokenAmount,
9                  uint64(block.timestamp));
10     vm.stopPrank();
11
12     //user wants to sell WETH tokens
13     uint256 wethToBuy = 1e18;
14     //poolTokens expected to receive
15     //since the pool ratio is 1:1, the user is expected to spend ~
16     //1 pool token
17     uint256 expectedPoolTokensToAdd = pool.
18         getInputAmountBasedOnOutput(wethToBuy, poolToken.balanceOf(
19             address(pool)), weth.balanceOf(address(pool)));
20     console.log("Expected PoolTokens to add:",
21         expectedPoolTokensToAdd);
22     uint256 inputReserve = poolToken.balanceOf(address(pool));
23     uint256 outputReserve = weth.balanceOf(address(pool));
24     uint256 outputAmount = wethToBuy;
25
26     uint256 actualPoolTokensToAdd = (1000 * (inputReserve *
27         outputAmount)) / (997 * (outputReserve - outputAmount));
28     console.log("Actual PoolTokens to add:", actualPoolTokensToAdd);
29
30     ;
31
32     assertLt(actualPoolTokensToAdd, expectedPoolTokensToAdd, "User
33         is charged more than intended");
34
35     //initiate the swap
36     address someUser = makeAddr("someUser");
37     poolToken.mint(someUser, 11 ether); //minting more than
38     //expected to ensure the user has enough balance
39     vm.startPrank(someUser);
40     poolToken.approve(address(pool), type(uint256).max);
41     pool.swapExactOutput(poolToken, weth, wethToBuy, uint64(block.
42         timestamp));
43     vm.stopPrank();
44
45     uint256 poolTokensAfterSwap = poolToken.balanceOf(someUser);
46     //after swap the user is left with less than 1 pool token (i.e
47     //spent nearly 10 times of the intended ~10 pool tokens), when
48     //they should have spent nearly 1 pool token for the purchase
49     //of 1 WETH
50     assertLt(poolTokensAfterSwap, 1e18, "User should have spent
51         less than 1 pool token");
52
53 }
```

Recommended Mitigation:

```
1  function getInputAmountBasedOnOutput(
2      uint256 outputAmount,
3      uint256 inputReserves,
```

```
4      uint256 outputReserves
5    )
6    public
7    pure
8    revertIfZero(outputAmount)
9    revertIfZero(outputReserves)
10   returns (uint256 inputAmount)
11   {
12 -     return ((inputReserves * outputAmount) * 10000) / ((
13 +     return ((inputReserves * outputAmount) * 1000) / ((
14   }
```

[H-2] Lack of slippage protection in TSwapPool::swapExactOutput causes users to spend way more tokens

Description: The `swapExactOutput` function does not include slippage protection, which means that users can end up spending significantly more tokens than they expect. The function calculates the input amount based on the output amount and reserves, but it does not check if the input amount exceeds a certain threshold.

This function is similar to `TSwapPool::swapExactInput` where the function specifies a `minOutputAmount`, the `swapExactOutput` function should specify a `maxInputAmount` to protect against slippage.

Impact: If the market conditions change between the time the user initiates the swap and the time it is executed, the user may end up spending much more than they intended, leading to potential loss of funds.

Proof of Concept:

1. The price of 1 WETH is 1 USDC.
2. The user wants to buy 1 WETH using pool tokens.
3. User inputs the `swapExactOutput` looking to buy 1 WETH, but does not specify a maximum amount of pool tokens they are willing to spend.
 - inputToken = poolToken
 - outputToken = weth
 - outputAmount = 1WETH
 - deadline = whatever
4. As the transaction is waiting in the mempool, the market changes.

- Hence, the user ends up spending nearly 10 pool tokens for 1 WETH as the function does not have slippage protection.

```
1  function test_No_Slippage_Protection_In_SwapExactOutput() external {
2      vm.startPrank(liquidityProvider);
3      weth.approve(address(pool), 100e18);
4      poolToken.approve(address(pool), 100e18);
5      pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6      vm.stopPrank();
7
8      uint256 wethToBuy = 1e18;
9      //user doesn't want to spend more than 2e18 pool tokens for 1
        e18 WETH
10     //since the pool ratio is 1:1, the user is expected to spend ~
        1 pool token
11     //initiate the swap
12     address someUser = makeAddr("someUser");
13     poolToken.mint(someUser, 11 ether); //minting more than
        expected to ensure the user has enough balance
14     vm.startPrank(someUser);
15     poolToken.approve(address(pool), type(uint256).max);
16     pool.swapExactOutput(poolToken, weth, wethToBuy, uint64(block.
        timestamp));
17     vm.stopPrank();
18
19     uint256 poolTokensAfterSwap = poolToken.balanceOf(someUser);
20     //after swap the user is left with less than 1 pool token due
        to price slippage, when they should have spent nearly 1 pool
        token for the purchase of 1 WETH
21     assertLt(poolTokensAfterSwap, 1e18, "User should have spent
        less than 1 pool token");
22 }
```

Recommended Mitigation: We should include a `maxInputAmount` parameter in the `swapExactOutput` function to allow users to set a limit on the maximum amount of input tokens they are willing to spend.

```
1  function swapExactOutput(
2      IERC20 inputToken,
3      IERC20 outputToken,
4      uint256 outputAmount,
5  +   uint256 maxInputAmount,
6      uint64 deadline
7  )
8      public
9      revertIfZero(outputAmount)
10     revertIfDeadlinePassed(deadline)
11     returns (uint256 inputAmount)
12 {
13     uint256 inputReserves = inputToken.balanceOf(address(this));
```

```
14         uint256 outputReserves = outputToken.balanceOf(address(this));
15
16         inputAmount = getInputAmountBasedOnOutput(outputAmount,
17             inputReserves, outputReserves);
17 +         if (inputAmount > maxInputAmount) {
18 +             revert TSwapPool__InputTooHigh(inputAmount, maxInputAmount)
19 +         ;
19 +     }
20
21
22         _swap(inputToken, inputAmount, outputToken, outputAmount);
23     }
```

[H-3] TSwapPool::sellPoolTokens mismatches input and output tokens causing users to receive incorrect amount of tokens

Description: The `sellPoolTokens` function in the `TSwapPool` contract is designed to allow users to sell their pool tokens for a specified output token. Users indicate how many pool tokens they are willing to sell in the `poolTokenAmount` parameter. However, the function miscalculates the swapped amount.

This is due to the fact that `swapExactOutput` function is called whereas this should be `swapExactInput` called rather because users specify the exact amount of input tokens not output.

Impact: Users will swap the wrong amount of tokens, which is a severe disruption of the protocol's functionality.

Proof of Concept:

1. User has 10 WETH and 10 Pool tokens.
2. User wants to 10 Pool tokens to get ~9 WETH, totally to ~19 WETH in user's balance
3. Rather function tries to take these 10 Pool tokens as output tokens and expects user to have ~11 WETH for the transaction to go through.
4. Hence, reverts while performing the unintended operation.

```
1  function test_SellPoolTokens_Reverts() external {
2      uint256 initialWethAmount = 100e18;
3      uint256 initialPoolTokenAmount = 100e18;
4      //initial liquidity added to the pool
5      vm.startPrank(LiquidityProvider);
6      weth.approve(address(pool), type(uint256).max);
7      poolToken.approve(address(pool), type(uint256).max);
8      pool.deposit(initialWethAmount, 0, initialPoolTokenAmount,
          uint64(block.timestamp));
```

```
9         vm.stopPrank();
10
11         //user wants to sell pool tokens
12         uint256 poolTokenToSell = 10e18;
13         //weth expected to receive
14         uint256 expectedWethToReceive = pool.
            getOutputAmountBasedOnInput(poolTokenToSell,
15             poolToken.balanceOf(address(pool)), weth.balanceOf(address(
                pool)));
16         console.log("Expected WETH to receive:", expectedWethToReceive)
            ;
17
18         vm.startPrank(user);
19         poolToken.approve(address(pool), type(uint256).max);
20
21         vm.expectPartialRevert(IERC20Errors.ERC20InsufficientBalance.
            selector);
22
23         pool.sellPoolTokens(poolTokenToSell);
24         vm.stopPrank();
25     }
```

Recommended Mitigation: Consider changing the implementation to use `swapExactInput` rather than `swapExactOutput`. This would also require changing the `sellPoolTokens` function to accept a new parameter `minWethToReceive` to be passed to `swapExactInput`.

```
1     function sellPoolTokens(uint256 poolTokenAmount, uint256
        minWethToReceive) external returns (uint256 wethAmount) {
2 -         return swapExactOutput(i_poolToken, i_wethToken,
        poolTokenAmount, uint64(block.timestamp));
3 +         return swapExactInput(i_poolToken, poolTokenAmount,
        i_wethToken, minWethToReceive)
4     }
```

It might be wise to add a deadline to the function, as currently there is no deadline to curb MEV.

[H-4] In `TSwapPool : : _swap` the extra tokens given to users after every `swapCount` reaches 10 or more, breaks the invariant of $x * y = k$

Description: The protocol follows a strict invariant of $x * y = k$, where:

- x : balance of pool token
- y : balance of weth
- k : constant product of two balances

This means whenever the balances change in the protocol, the ratio between the two amounts should remain constant, hence the k . However, this is broken due to extra incentive in the `_swap` function.

Meaning that overtime the protocol funds will be drained.

```
1 //The following block of code is repesible for the issue
2 swap_count++;
3 if (swap_count >= SWAP_COUNT_MAX) {
4     swap_count = 0;
5     outputToken.safeTransfer(msg.sender, 1_000_000_000_000_000_000);
6 }
```

Impact: A user could maliciously drain the protocol funds by doing a lot of swaps and collecting the extra incentive given out by the protocol.

More simply put, the protocol's core invariant is broken.

Proof of Concept:

1. A user swaps 10 times, and collects the extra incentive of 1_000_000_000_000_000_000 tokens.
2. The user continues to swap until all the protocol funds are drained.

Proof of Code

Place the following into TSwapPool.t

```
1     function test_invariant_broke_x_product_y_not_constant() external {
2
3         vm.startPrank(liquidityProvider);
4         weth.approve(address(pool), 100e18);
5         poolToken.approve(address(pool), 100e18);
6         pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
7         vm.stopPrank();
8
9         uint256 outputWeth = 1e17;
10        poolToken.mint(user, 100e18);
11
12        vm.startPrank(user);
13        poolToken.approve(address(pool), type(uint256).max);
14        pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
15        pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
16        pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
17        pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
18        pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
19        pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
20        pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.timestamp));
21    }
```

```
21     pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.  
22         timestamp));  
23     pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.  
24         timestamp));  
25     int256 startingY = int256(weth.balanceOf(address(pool)));  
26     int256 expectedDeltaY = int256(outputWeth) * int256(-1);  
27     pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.  
28         timestamp));  
29     vm.stopPrank();  
30  
31     int256 endingY = int256(weth.balanceOf(address(pool)));  
32  
33     int256 actualDeltaY = endingY - startingY;  
34     assertEq(expectedDeltaY, actualDeltaY);  
35  
36 }
```

Recommended Mitigation: Remove the extra incentive mechanism. If you want to keep this in, we should account for the change in the $x * y = k$ protocol invariant or we should set aside the tokens in the same way we do for fees.

```
1 -     swap_count++;  
2 -     if (swap_count >= SWAP_COUNT_MAX) {  
3 -         swap_count = 0;  
4 -         outputToken.safeTransfer(msg.sender, 1_000_000_000_000_000_000)  
5 -     ;  
6 - }
```

Medium

[M-1] TSwapPool::deposit is missing a deadline parameter check causing transactions to be processed after the deadline

Description: The `TSwapPool::deposit` function in the `TSwapPool` contract includes a `deadline` parameter which according to the documentation is **The deadline for the transaction to be completed by**, but there is no check to ensure that the transaction is processed before the deadline. This could allow users to execute **add liquidity** transactions at an unexpected time, leading to potential loss of funds when the deposit rate is unfavorable.

Impact: Transactions could be sent when market conditions are unfavourable to deposit even after adding a deadline parameter to the function without checking it could lead to unexpected behavior and potential loss of funds.

Proof of Concept: The `deadline` parameter is unused.

1. Producing Sandwich attack on deposit function causing user to receive less tokens than intended

Sandwich attack POC

```
1  function test_SandwichAttackOnDeposit() external {
2      //initial liquidity added to the pool:
3      vm.startPrank(liquidityProvider);
4      weth.approve(address(pool), 100e18);
5      poolToken.approve(address(pool), 100e18);
6      pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
7      vm.stopPrank();
8
9      //user tries to make a deposit of 5e18 WETH
10     uint256 wethAmountToDeposit = 5e18;
11     uint256 poolTokenToDepositBeforeFrontRun = pool.
        getPoolTokensToDepositBasedOnWeth(wethAmountToDeposit);
12
13     //attacker tries to front-run the deposit by making a swap
        trade
14     uint256 outputWeth = 7e18;
15     uint256 inputReserve_PoolToken = poolToken.balanceOf(address(
        pool));
16     uint256 outputReserve_Weth = weth.balanceOf(address(pool));
17     uint256 inputPoolToken =
18         pool.getInputAmountBasedOnOutput(outputWeth,
            inputReserve_PoolToken, outputReserve_Weth);
19     // Record attacker's initial balances
20     uint256 attackerWethBefore = weth.balanceOf(attacker);
21     uint256 attackerPoolTokenBefore = poolToken.balanceOf(attacker)
        ;
22
23     poolToken.mint(attacker, inputPoolToken + 1); //just to be sure
24
25     vm.startPrank(attacker);
26     poolToken.approve(address(pool), type(uint256).max);
27     pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.
        timestamp));
28     vm.stopPrank();
29
30     //user makes a deposit
31     uint256 poolTokenToDepositAfterFrontRun = pool.
        getPoolTokensToDepositBasedOnWeth(wethAmountToDeposit);
32     assertGt(poolTokenToDepositAfterFrontRun,
        poolTokenToDepositBeforeFrontRun, "Front-run failed");
33
34     vm.startPrank(user);
35     require(poolToken.balanceOf(user) >=
        poolTokenToDepositAfterFrontRun, "Not enough pool tokens");
36     poolToken.approve(address(pool), type(uint256).max);
```

```
37     weth.approve(address(pool), type(uint256).max);
38     pool.deposit(wethAmountToDeposit, 0,
39         poolTokenToDepositAfterFrontRun, uint64(block.timestamp));
40     vm.stopPrank();
41     // BACKRUN: Attacker reverses the trade (WETH -> PoolToken)
42     vm.startPrank(attacker);
43     uint256 wethToSwapBack = outputWeth;
44     weth.approve(address(pool), wethToSwapBack);
45     pool.swapExactInput(weth, wethToSwapBack, poolToken, 0, uint64(
46         block.timestamp));
47     vm.stopPrank();
48     // Calculate attacker's profit
49     uint256 attackerWethAfter = weth.balanceOf(attacker);
50     uint256 attackerPoolTokenAfter = poolToken.balanceOf(attacker);
51
52     // Attacker should have more PoolToken than they started with
53     int256 wethProfit = int256(attackerWethAfter) - int256(
54         attackerWethBefore);
55     int256 poolTokenProfit = int256(attackerPoolTokenAfter) -
56         int256(attackerPoolTokenBefore);
57
58     console.log("Attacker WETH profit:", wethProfit);
59     console.log("Attacker PoolToken profit:", poolTokenProfit);
60     console.log("User paid extra PoolToken:",
61         poolTokenToDepositAfterFrontRun -
62         poolTokenToDepositBeforeFrontRun);
63
64     // The attacker should be profitable in at least one token
65     assertTrue(poolTokenProfit > 0 || wethProfit > 0, "Attack
66         should be profitable");
67 }
```

Recommended Mitigation: Consider making the following changes to the function.

```
1     function deposit(
2         uint256 wethToDeposit,
3         uint256 minimumLiquidityTokensToMint,
4         uint256 maximumPoolTokensToDeposit,
5         uint64 deadline
6     )
7     external
8     revertIfZero(wethToDeposit)
9 +   revertIfDeadlinePassed(deadline)
10    returns (uint256 liquidityTokensToMint){
11    .
12    .
13    .
14 }
```

[M-2] Rebase, fee-on-transfer and ERC-777 break protocol invariant $x * y = k$

Description: The TSwap AMM pool fails to account for various non-standard ERC20 token behaviors, causing systematic violations of the core AMM invariant $x * y = k$.

The pool's accounting system assumes standard ERC20 transfer behavior where the exact amount specified in `transferFrom()` is received by the pool.

However, multiple categories of **weird ERC20** tokens such as fee-on-transfer, rebase, reentrant, pausable tokens, etc can cause discrepancies between expected and actual token balances, breaking the fundamental mathematical relationship that ensures proper price discovery, swap calculations, and liquidity management. For example -

Fee-on-Transfer Tokens:

- Pool expects: `transferFrom(user, pool, 100e18)` → pool receives 100e18
- Reality: Pool receives 100e18 - fee (e.g., 99e18)

Impact: $x * y = k$ breaks, damaging the integrity of the AMM model.

Proof of Concept:

1. A user adds liquidity for a (weird ERC20 - fee on transfer) token.
2. Pool receives less than the intended liquidity for the given amount of WETH.
3. Hence, the invariants breaks.

Proof of Code

Place the following into TSwapPool.t

```
1 function test_invariant_broken_for_Weird_ERC20() external {
2     TransferFeeToken newPoolToken = new TransferFeeToken(1e15, "
3         Weird ERC20", "wERC20");
4
5     weth = new ERC20Mock();
6     pool = new TSwapPool(address(newPoolToken), address(weth), "
7         LTokenA", "LA");
8
9     weth.mint(LiquidityProvider, 200e18);
10    newPoolToken.mint(LiquidityProvider, 200e18);
11
12    weth.mint(user, 10e18);
13    newPoolToken.mint(user, 10e18);
14
15    vm.startPrank(LiquidityProvider);
16    weth.approve(address(pool), 100e18);
17    newPoolToken.approve(address(pool), 100e18);
18    pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
```



```
17         vm.stopPrank();
18
19         uint256 wethAmount = 5e18;
20         int256 startingX = int256(newPoolToken.balanceOf(address(pool))
21             );
22         uint256 poolTokenAmount = pool.
23             getPoolTokensToDepositBasedOnWeth(wethAmount);
24         int256 expectedDeltaX = int256(poolTokenAmount);
25
26         vm.startPrank(user);
27         weth.approve(address(pool), wethAmount);
28         newPoolToken.approve(address(pool), poolTokenAmount);
29         pool.deposit(wethAmount, 0, poolTokenAmount, uint64(block.
30             timestamp));
31         vm.stopPrank();
32
33         int256 endingX = int256(poolToken.balanceOf(address(pool)));
34         int256 actualDeltaX = endingX - startingX;
35
36         assertEq(expectedDeltaX, actualDeltaX, "Delta X mismatch");
37     }
38 }
```

Recommended Mitigation: - Use balance delta checks before/after transferFrom and transfer to conform the integrity of AMM. - Only allow a known set of vetted tokens into pools, explicitly excluding known non-standard ones

Low

[L-1] The arguments set for this event TSwapPool::LiquidityAdded are not set in correct order

Description: The event `LiquidityAdded` is emitting the wrong arguments from function `TSwapPool::_addLiquidityMintAndTransfer`. The order of the arguments should be `wethDeposited` first, then `poolTokensDeposited`.

```
1 //event definition
2 event LiquidityAdded(address indexed liquidityProvider, uint256
3     wethDeposited, uint256 poolTokensDeposited);
```

```
1 //event emission
2 emit LiquidityAdded(msg.sender, poolTokensToDeposit, wethToDeposit);
```

Impact: This could lead to confusion for developers reading the code, as it suggests that there is a specific error condition that can occur, but it is not actually being checked for or handled anywhere in the contract.

Recommended Mitigation: Change the order of the arguments in the event definition or the event emission to match the correct order.

```
1 - emit LiquidityAdded(msg.sender, poolTokensDeposited, wethDeposited);
2 + emit LiquidityAdded(msg.sender, wethDeposited, poolTokensDeposited);
```

[L-2] Default value returned by TSwapPool::swapExactInput results in incorrect value being returned

Description: The `swapExactInput` is expected to return the amount of output tokens received after the swap, but it currently returns a default value of 0. However, while it declares the named returned value `output`, it is never assigned a value, not uses an explicit `return` statement.

Impact: The return value will always be 0, giving incorrect information to the caller.

Proof of Concept:

```
1 function test_SwapExactInput_Always_Returns_Zero() external {
2     vm.startPrank(liquidityProvider);
3     weth.approve(address(pool), 100e18);
4     poolToken.approve(address(pool), 100e18);
5     pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6     vm.stopPrank();
7
8     vm.startPrank(user);
9     poolToken.approve(address(pool), 10e18);
10    // After we swap, there will be ~110 tokenA, and ~91 WETH
11    // 100 * 100 = 10,000
12    // 110 * ~91 = 10,000
13    uint256 expected = 9e18;
14
15    uint256 actualAmount = pool.swapExactInput(poolToken, 10e18,
16        weth, expected, uint64(block.timestamp));
17    vm.stopPrank();
18    assert(weth.balanceOf(user) >= expected);
19    assertEq(actualAmount, 0, "SwapExactInput should always return 0");
20 }
```

Recommended Mitigation:

```
1 function swapExactInput(
2     IERC20 inputToken,
3     uint256 inputAmount,
4     IERC20 outputToken,
```

```
5      uint256 minOutputAmount,  
6      uint64 deadline  
7  )  
8      public  
9      revertIfZero(inputAmount)  
10     revertIfDeadlinePassed(deadline)  
11 -     returns (uint256 output)  
12 +     returns (uint256 outputAmount)  
13     {  
14         uint256 inputReserves = inputToken.balanceOf(address(this));  
15         uint256 outputReserves = outputToken.balanceOf(address(this));  
16  
17 -         uint256 outputAmount = getOutputAmountBasedOnInput(inputAmount  
18 +         , inputReserves, outputReserves);  
19         outputAmount = getOutputAmountBasedOnInput(inputAmount,  
20         inputReserves, outputReserves);  
21  
22         if (outputAmount < minOutputAmount) {  
23             revert TSwapPool__OutputTooLow(outputAmount,  
24             minOutputAmount);  
25         }  
26  
27         _swap(inputToken, inputAmount, outputToken, outputAmount);  
28     }
```

Informational

[I-1] Unused custom error `PoolFactory::PoolDoesNotExist` in `PoolFactory` contract

Description: The custom error `PoolFactory__PoolDoesNotExist` is defined in the `PoolFactory` contract but is never used.

Impact: This could lead to confusion for developers reading the code, as it suggests that there is a specific error condition that can occur, but it is not actually being checked for or handled anywhere in the contract.

Recommended Mitigation: Remove the unused error or implement checks for the error condition in the contract.

```
1 -     error PoolFactory__PoolDoesNotExist();
```

[I-2] wethToken parameter in the constructor of the PoolFactory contract is not checked for zero address

Description: The `wethToken` parameter in the constructor is not validated to ensure it is not a zero address.

Impact: If a zero address is passed as the WETH token address, it could lead to unexpected behavior or vulnerabilities in the contract.

Recommended Mitigation: Add a check in the constructor to ensure that the `wethToken` address is not the zero address.

```
1 +         require(wethToken != address(0), "WETH token address cannot be  
    zero");
```

[I-3] liquidityTokenSymbol variable of PoolFactory::createPool function should use symbol() instead of name()

Description: The `liquidityTokenSymbol` variable is constructed using the `name()` function of the ERC20 token, but it should use the `symbol()` function instead to create a proper token symbol.

Impact: Using the token name instead of the symbol could lead to incorrect or unexpected token symbols being generated for the liquidity tokens.

Recommended Mitigation: Update the `liquidityTokenSymbol` variable to use the `symbol()` function of the ERC20 token instead of the `name()` function.

```
1 -         string memory liquidityTokenSymbol = string.concat("ts",  
    IERC20(tokenAddress).name());  
2 +         string memory liquidityTokenSymbol = string.concat("ts",  
    IERC20(tokenAddress).symbol());
```

[I-4] deadline parameter in TSwapPool::deposit function is not used

Description: The `deadline` parameter in the `deposit` function of the `TSwapPool` contract is not utilized within the function body.

Impact: Having unused parameters can lead to confusion and may indicate incomplete functionality. It could also potentially waste gas if the parameter is not needed.

Recommended Mitigation: Implement the logic in `deposit` function to utilize `deadline` parameter effectively.

```

1      function deposit(
2          uint256 wethToDeposit,
3          uint256 minimumLiquidityTokensToMint,
4          uint256 maximumPoolTokensToDeposit,
5 +      uint64 deadline
6      )
7          external
8          revertIfZero(wethToDeposit)
9 +      revertIfDeadlinePassed(deadline)
10         returns (uint256 liquidityTokensToMint){
11     .
12     .
13     .
14     }

```

[I-5]: Event is missing indexed fields

Index event fields make the field more quickly accessible to off-chain tools that parse events. However, note that each index field costs extra gas during emission, so it's not necessarily best to index the maximum allowed per event (three fields). Each event should use three indexed fields if there are three or more fields, and gas usage is not particularly of concern for the events in question. If there are fewer than three fields, all of the fields should be indexed.

```

1 -      event PoolCreated(address tokenAddress, address
2 +      poolAddress);
2 +      event PoolCreated(address indexed tokenAddress, address
      indexed poolAddress);

```

```

1 -      event LiquidityAdded(address indexed liquidityProvider,
2 +      uint256 wethDeposited, uint256 poolTokensDeposited);
2 +      event LiquidityAdded(address indexed liquidityProvider,
      uint256 indexed wethDeposited, uint256 indexed poolTokensDeposited);

```

```

1 -      event LiquidityRemoved(address indexed liquidityProvider,
2 +      uint256 wethWithdrawn, uint256 poolTokensWithdrawn);
2 +      event LiquidityRemoved(address indexed liquidityProvider,
      uint256 indexed wethWithdrawn, uint256 indexed poolTokensWithdrawn);

```

```

1 -      event Swap(address indexed swapper, IERC20 tokenIn,
2 +      uint256 amountTokenIn, IERC20 tokenOut, uint256 amountTokenOut);
2 +      event Swap(address indexed swapper, IERC20 indexed tokenIn
      , uint256 amountTokenIn, IERC20 indexed tokenOut, uint256
      amountTokenOut);

```

4 Found Instances

- Found in src/PoolFactory.sol Line: 35

```
1          event PoolCreated(address tokenAddress, address poolAddress);
```

- Found in src/TSwapPool.sol Line: 43

```
1          event LiquidityAdded(address indexed liquidityProvider, uint256
            wethDeposited, uint256 poolTokensDeposited);
```

- Found in src/TSwapPool.sol Line: 44

```
1          event LiquidityRemoved(address indexed liquidityProvider,
            uint256 wethWithdrawn, uint256 poolTokensWithdrawn);
```

- Found in src/TSwapPool.sol Line: 45

```
1          event Swap(address indexed swapper, IERC20 tokenIn, uint256
            amountTokenIn, IERC20 tokenOut, uint256 amountTokenOut);
```

[I-6] TSwapPool::TSwapPool__WethDepositAmountTooLow in this event, TSwapPool::MINIMUM_WETH_LIQUIDITY is not required to be emitted as this a constant value

Description: The event `TSwapPool::TSwapPool__WethDepositAmountTooLow` emits the constant value `TSwapPool::MINIMUM_WETH_LIQUIDITY`, which is unnecessary.

Impact: Unnecessary emissions can lead to increased gas costs and cluttered event logs, making it harder for off-chain tools to parse relevant information.

Recommended Mitigation: Remove the emission of `TSwapPool::MINIMUM_WETH_LIQUIDITY` from the event `TSwapPool::TSwapPool__WethDepositAmountTooLow`.

```
1  -          error TSwapPool__WethDepositAmountTooLow(uint256
            minimumWethDeposit, uint256 wethToDeposit);
2  +          error TSwapPool__WethDepositAmountTooLow(uint256 wethToDeposit
            );
```

```
1  -          revert TSwapPool__WethDepositAmountTooLow(
            MINIMUM_WETH_LIQUIDITY, wethToDeposit);
2  +          revert TSwapPool__WethDepositAmountTooLow(wethToDeposit);
```

[I-7] TSwapPool::deposit function is not following CEI for the first-time liquidity deposit

Description: The `TSwapPool::deposit` function does not follow the Checks-Effects-Interactions (CEI) pattern for the first-time liquidity deposit. Specifically, it performs state changes (effects) after interactions, which can lead to unexpected behavior and vulnerabilities.

Impact: This can lead to reentrancy attacks or other unexpected behaviors, especially if the contract interacts with external contracts or tokens.

Recommended Mitigation: Ensure that the function follows the CEI pattern by performing all checks first, then making state changes, and finally interacting with external contracts or tokens.

```
1      else {  
2  +    liquidityTokensToMint = wethToDeposit;  
3      _addLiquidityMintAndTransfer(wethToDeposit,  
        maximumPoolTokensToDeposit, wethToDeposit);  
4  -    liquidityTokensToMint = wethToDeposit;  
5      }
```

[I-8] Usage of magic numbers such as 997 and 1000 is not recommended

Description: Magic numbers are hard-coded values that appear in code without explanation. They can make code difficult to understand and maintain.

```
1      uint256 inputAmountMinusFee = inputAmount * 997;  
2      uint256 denominator = (inputReserves * 1000) +  
        inputAmountMinusFee;
```

Impact: Using magic numbers can lead to confusion and errors, as the meaning of these numbers is not immediately clear. It can also make future modifications more difficult.

Recommended Mitigation: Define constants for these magic numbers to improve code readability and maintainability.

```
1  +    uint256 constant FEE_DENOMINATOR = 1000;  
2  +    uint256 constant FEE_MULTIPLIER = 997;  
3      uint256 inputAmountMinusFee = inputAmount * FEE_MULTIPLIER;  
4      uint256 denominator = (inputReserves * FEE_DENOMINATOR) +  
        inputAmountMinusFee;
```

[I-9] Natspec is not defined for the function TSwapPool::swapExactInput

Description: The function `TSwapPool::swapExactInput` does not have Natspec comments defined, which are used to provide documentation for functions, parameters, and return values.

Impact: Without Natspec comments, it is difficult for developers to understand the purpose and usage of the function, which can lead to misuse or errors.

Recommended Mitigation: Add Natspec comments to the function to provide clear documentation for its purpose, parameters, and return values.

```
1 +    /// @notice Swaps an exact amount of input tokens for as many
    output tokens as possible
2 +    /// @param inputToken The address of the input token
3 +    /// @param inputAmount The exact amount of input tokens to swap
4 +    /// @param outputToken The address of the output token
5 +    /// @param minOutputAmount The minimum amount of output tokens to
    receive
6 +    /// @param deadline The deadline for the transaction to be
    completed by
7 +    /// @return amountOut The amount of output tokens received
8      function swapExactInput(
9          IERC20 inputToken,
10         uint256 inputAmount,
11         IERC20 outputToken,
12         uint256 minOutputAmount,
13         uint64 deadline
14     ) external returns (uint256 amountOut) {
15         // Swap logic here
16     }
```

[I-10] TSwapPool::swapExactOutput natspec is missing the @param tag for deadline

Description: The @param tag for `deadline` is missing in the Natspec comments for the `TSwapPool::swapExactOutput` function.

Impact: Without the @param tag, users may not understand the purpose of the `deadline` parameter, leading to potential misuse or errors.

Recommended Mitigation: Add the @param tag for `deadline` in the Natspec comments for the `TSwapPool::swapExactOutput` function.

```
1 +    /// @param deadline The deadline for the transaction to be
    completed by
2      /// @return amountIn The amount of input tokens used for the swap
3      function swapExactOutput(
4          IERC20 inputToken,
5          uint256 outputAmount,
6          IERC20 outputToken,
7          uint256 maxInputAmount,
8          uint64 deadline
9      )
10     external
```



```
11     revertIfZero(outputAmount)
12     revertIfDeadlinePassed(deadline)
13     returns (uint256 amountIn)
14     {      // Swap logic here
15     }
```

Gas

[G-1] TSwapPool::swapExactInput and TSwapPool::totalLiquidityTokenSupply functions should be marked as external instead of public

Description: The `TSwapPool::swapExactInput` and `TSwapPool::totalLiquidityTokenSupply` functions are currently marked as **public**, but they should be marked as **external** to optimize gas usage and restrict access to external calls only.

Impact: Marking the function as **external** can reduce gas costs by allowing the Solidity compiler to optimize the function call.

Recommended Mitigation: Change the visibility of the `swapExactInput` function from **public** to **external**.

```
1     function swapExactInput(
2         IERC20 inputToken,
3         uint256 inputAmount,
4         IERC20 outputToken,
5         uint256 minOutputAmount,
6         uint64 deadline
7     )
8 -     public returns (uint256 amountOut) {
9 +     external returns (uint256 amountOut) {
10         // Swap logic here
11     }
```

Change the visibility of the `totalLiquidityTokenSupply` function from **public** to **external**.

```
1     function totalLiquidityTokenSupply()
2 -     public view returns (uint256) {
3 +     external view returns (uint256) {
4         return totalSupply();
5     }
```